THE UNIVERSITY OF SYDNEY

MATH2008 Introduction to Modern Algebra

(http://www.maths.usyd.edu.au/u/UG/IM/MATH2008/)

Semester 2, 2003 — Lecturer: R. Howlett

Computer Tutorial 11

1. Define `G := Sym(9)`, and choose any permutations x and y that move only one number in common. For example `x := G!(1,4,5,6)` and `y := G!(5,7,8,9)` would do.

(*i*) Use MAGMA to compute the permutation $x^{-1}y^{-1}xy$. (You may either type this as it stands or use the MAGMA abbreviation `(x,y)` for the element `x^(-1)*y^(-1)*x*y`.)

(*ii*) Repeat this for several other choices of x and y. What do you observe about the result? Try calculating some of the products by hand to see if you can find a reason for what you observe.

Solution.

```
> G:=Sym(9);
> x:=G!(1,2,3)(5,7);
> y:=G!(4,5,8,9);
> (x,y);
(5, 8, 7)
> z:=G!(3,4,6)(8,9);
> (x,z);
(1, 3, 4)
```

```
> (z,x);
(1, 4, 3)
> w:=G!(5,8,1,7,2);
> (z,w);
(1, 9, 8)
> (w,z);
(1, 8, 9)
```

The result is always a 3-cycle. Let i be the number that is moved by both x and y, and let $j = i^{x^{-1}}$ and $k = i^x$. Thus j and k are the numbers that appear on either side of i in the expression for x. For example, if $x = (1,4,5,6)$ and $y = (5,7,8,9)$ then $j = 4$, $i = 5$ and $k = 6$. Similarly, let $l = i^{y^{-1}}$ and $m = i^y$. In our example we would have $l = 9$ and $m = 7$. It turns out that $x^{-1}y^{-1}xy$ is actually the 3-cycle (i, m, k).

As a first step to seeing this, observe that as i is the only number that both x and y move, y fixes j and k (since since x does not fix these two) and x fixes l and m (since y does not). Now consider what $x^{-1}y^{-1}xy$ does to i. Starting from i, apply successively x^{-1}, y^{-1}, x and y:

$$i \xrightarrow{x^{-1}} j \xrightarrow{y^{-1}} j \xrightarrow{x} i \xrightarrow{y} m.$$

Now consider what $x^{-1}y^{-1}xy$ does to m:

$$m \xrightarrow{x^{-1}} m \xrightarrow{y^{-1}} i \xrightarrow{x} k \xrightarrow{y} k.$$

Finally, consider what $x^{-1}y^{-1}xy$ does to k:

$$k \xrightarrow{x^{-1}} i \xrightarrow{y^{-1}} l \xrightarrow{x} l \xrightarrow{y} i.$$

So (i, m, k) is one of the cycles appearing in $x^{-1}y^{-1}xy$. It remains to show that $x^{-1}y^{-1}xy$ fixes everything else.

Choose any number n that is not one of i, m or k. If x and y both fix n then it is clear that $x^{-1}y^{-1}xy$ also fixes n. Now suppose that x moves n, and put $p = n^{x^{-1}}$. Since $n \neq k$, we know that $p \neq k^{x^{-1}} = i$. So neither p nor n is equal to i, and since x moves both p and n it follows that y does not move either p or n. So, on applying $x^{-1}y^{-1}xy$, we find that

$$n \xrightarrow{x^{-1}} p \xrightarrow{y^{-1}} p \xrightarrow{x} n \xrightarrow{y} n.$$

That is, n is fixed by $x^{-1}y^{-1}xy$. Finally, suppose that y moves n, and put $p = n^{y^{-1}}$. Since $n \neq m$, we know that $p \neq m^{y^{-1}} = i$. So neither p nor n is equal to i, and since y moves both p and n it follows that x does not move either p or n. So, on applying $x^{-1}y^{-1}xy$, we find that

$$n \xrightarrow{x^{-1}} n \xrightarrow{y^{-1}} p \xrightarrow{x} p \xrightarrow{y} n.$$

So n is fixed by $x^{-1}y^{-1}xy$ in this case too, and therefore i, m and k are the only things moved by $x^{-1}y^{-1}xy$.

2. Use the following commands to set up subgroups H, K and L of Alt(5).

```
G := Alt(5);
H := Stabilizer(G,3);
K := Stabilizer(G,4);
L := Stabilizer(G,{3,4});
```

(*i*) Find the subgroup M which is the intersection of H and K. Is M a subgroup of L? (Use the MAGMA command `meet` to get the intersection.)

(*ii*) Is M equal to L? If not, explain why they differ, and how they are related.

Solution.

```
> G := Alt(5);
> H := Stabilizer(G,3);
> K := Stabilizer(G,4);
> L := Stabilizer(G,{3,4});
> L;
Permutation group L acting on a set of cardinality 5
Order = 6 = 2 * 3
(1, 2)(3, 4)
(2, 5)(3, 4)
> M := H meet K;
> print M subset L;
true
```

This shows that M is a subgroup of L.

```
> Index(L,M);
2
```

This shows that M has just two cosets in L. The number of elements in M is exactly half the number in L. The elements of L that are not in M interchange 3 and 4, rather than fixing them.

Of course, MAGMA can also tell us the order of M and elements that generate M.

```
> M;
Permutation group M acting on a set of cardinality 5
Order = 3
(1, 5, 2)
```

3. (*i*) Find a set of 3-cycles that generate the alternating group Alt(5). To do this you can set `A := Alt(5)` and then check various subgroups of the form
`sub< A | (1,2,3), ... >`
Find a generating set which is as small as possible.

(*ii*) Repeat Part (*i*) for Alt(6).

Solution.

```
> A:=Alt(5);
> #A;
60
> x:=A!(1,2,3);
> y:=A!(1,2,4);
> z:=A!(1,2,5);
> u:=A!(1,3,4);
> v:=A!(1,3,5);
> w:=A!(1,4,5);
> #sub<A|x,y,z,u,v,w>;
60
```

```
> #sub<A|x,y>;
12
> #sub<A|x,z>;
12
> #sub<A|x,u>;
12
> #sub<A|x,v>;
12
> #sub<A|x,w>;
60
```

Why do x and w generate Alt(5) while x and y do not? The point is that x and y both fix 5, and so the subgroup generated by x and y is contained in the stabilizer of 5 (which is a subgroup of order 12, isomorphic to Alt(4)). Similarly, x and z both fix 4, and hence cannot generate Alt(5). Similar observations hold for the pairs x, u and x, v. But there is no number that is fixed by both x and w.

In view of the above remarks, if we want a set of 3-cycles that generates Alt(6), we had better make sure that between them they move all the numbers 1, 2, 3, 4, 5 and 6. So let us try $\{(1,2,3),(4,5,6)\}$:

```
> A:=Alt(6);
> #A;
360
> #sub<A|A!(1,2,3),A!(4,5,6)>;
9
```

That failed. It failed because $(1,2,3)$ and $(4,5,6)$ both in the setwise stabilizer of $\{1,2,3\}$ (as well as the setwise stabilizer of $\{4,5,6\}$). So we will need at least three 3-cycles to generate Alt(6):

```
> #sub<A|A!(1,2,3),A!(4,5,6),A!(1,2,4)>;
360
```

4. (*i*) Let G be the symmetric group Sym(5) and use MAGMA to construct the following subsets

```
K1:= {G | (1,2),(1,3),(2,3),(2,4),(2,5),(3,5),(4,5)};
D := { x*G!(1,3,4) : x in Stabilizer(G,1) };
K2:= Set(G) diff {x*y : x,y in D};
K3:= K1 join K2;
K4:= { G!(1,2,3)*x : x in Stabilizer(G,1) };
```

(*ii*) Find the number of elements in each of `K1, K2, K3` and `K4`.

(*iii*) Which of the these sets is a right coset of a subgroup of G? If it is a right coset, what is the subgroup?

(*iv*) The set `K4` is a left coset of `H := Stabilizer(G,1)`. In Part (*iii*) you will have discovered that it is also a right coset of some subgroup. Is it always true that every left coset of a subgroup H is also a right coset of some subgroup? Must the subgroups concerned always be equal?

Solution.

```
> G := Sym(5);
> #G;
120
> K1 := {G | (1,2),(1,3),(2,3),(2,4),(2,5),(3,5),(4,5)};
> D := { x*G!(1,3,4) : x in Stabilizer(G,1) };
> K2 := Set(G) diff {x*y : x,y in D};
> K3 := K1 join K2;
> K4 := { G!(1,2,3)*x : x in Stabilizer(G,1) };
> #K1, #K2, #K3, #K4;
7 24 30 24
```

The order of a subgroup of G has to be a divisor of the order of G (Lagrange's Theorem), and the number of elements in any coset of a subgroup has to be the same as the order of the subgroup itself. Since 7 is not a divisor of 120, $K1$ is certainly not a coset of any subgroup.

If H is a subgroup of G and x any element of G then the right coset of H containing x is the set $Hx = \{ hx \mid h \in H \}$. (It does contain x, since the identity element is in H.) Recall that distinct cosets have no elements in common. Now if y is any element of Hx then y is in both Hy and Hx, and so it follows that $Hy = Hx$. So if a subset K of G is a right coset of some subgroup H, then we can choose any element $y \in K$ and it will be true that $K = Hy$. And if $K = Hy$ then $H = Ky^{-1} = \{ kx^{-1} \mid k \in K \}$.

The MAGMA startup file for this course defines a function `isClosed` that can be used to test whether or not a set Ky^{-1} is closed under multiplication. If it is closed under multiplication then it is a subgroup of G, otherwise it is not. (See Exercise 5 of Tutorial 10.) Or you can look at the subgroup of G generated by the set Ky^{-1}: this will be equal to Ky^{-1} if Ky^{-1} is a subgroup of G, otherwise it will be bigger than Ky^{-1}.


```
> x:=Random(K2);
> x;
(1, 3, 4, 5)
> H:={k*x^(-1): k in K2};
> isClosed(H);
true
```

So `K2` is a right coset.

```
> y:=Random(K3);
> y;
(1, 3)
> L:={k*y^(-1): k in K3};
> M:=sub<G|L>;
> #M;
120
> z:=Random(K4);
> z;
(1, 4, 2, 3)
> N:={k*z^(-1): k in K4};
> P:=sub<G|M>;
> #P;
24
> Set(P) eq M;
true
```

So `K3` is not a right coset, while `K4` is a right coset.

Let Q be the stabilizer of 1. By definition, `K4` is the left coset $(1,2,3)Q$. According to MAGMA's calculations above, `K4` is a right coset of the subgroup `P`.

```
> Q:=Stabilizer(G,1);
> P eq Q;
false
```

So it is possible for a set to simultaneously be a left coset of one subgroup and a right coset of another.

We have seen that if y is any element of the set K, and if K is a right coset of a subgroup H, then $K = Hy$. If K is also a left coset of a subgroup L then we must also have $K = yL$. So we have $yL = Hy$, from which it follows that $L = y^{-1}Hy$. It is in fact true that if H is a subgroup of G and y any element of G then $y^{-1}Hy$ is a subgroup of G. It may or may not equal H.